Alma Mater Studiorum - Università di Bologna

SCUOLA DI INGEGNERIA

CORSO DI LAUREA IN INGEGNERIA MECCATRONICA

TESI DI LAUREA

in

Informatica Industriale

SVILUPPO DI UN'APPLICAZIONE ROBOTICA INDUSTRIALE

CANDIDATA RELATORE

Sofia Chiarini Chiar.mo Prof. Eugenio Faldella

CORRELATORE

Ing. Alan Sassi

Anno Accademico 2021/2022

Sessione II

$A\ mio\ nonno$

Indice

1	Intr	roduzione	1
	1.1	Obiettivi	1
	1.2	Organizzazione dei contenuti	1
2	Il so	oftware plc	2
	2.1	Lo Structured Control Language	2
	2.2	La configurazione hardware	3
	2.3	Il protocollo Profinet	3
	2.4	Lo scambio dati	3
	2.5	La struttura del programma	6
		2.5.1 Il blocco OB Main	6
		2.5.2 Il blocco FC Inputs	6
		2.5.3 Il blocco FC AlarmsHandler	7
		2.5.4 Il blocco FC UnitsHandler	8
		2.5.5 Il blocco FC StateHandler	14
		2.5.6 Il blocco FC RecipeHandler	15
		2.5.7 Il blocco FC Comunication	16
		2.5.8 Il blocco FC Outputs	19
3	Il so	oftware del braccio robotico	20
	3.1	Introduzione al linguaggio rapid	20
		3.1.1 Le routine	20
		3.1.2 I tipi di dati	21
		3.1.3 La istruzioni	21
	3.2	Il modulo CalibData	21
	3.3	Il modulo WorldZoneModule	23
	3.4	Il modulo ErrModule	24
	3.5	Il modulo ToolModule	25
	3.6	Il modulo VisionSysMoule	26
	3.7	Il modulo InterruptsModule	30
	3.8	Il modulo PickAndPlaceModule	32
	3.9	Il modulo PallettizeModule	35
	3.10	Il modulo PallettizeInterlockingModule	37
4	Il si	istema di visione	38
	4.1	Il software per il formato standard	38
	4.2	Il software per il formato bicolore	39
	43	Luminosità e calibrazione	30

5	L'in	terfaccia uomo-macchina	40
	5.1	La pagina Home	40
	5.2	La pagina di manutenzione	43
	5.3	La pagina delle segnalazioni	44
	5.4	La pagina delle ricette	45
	5.5	La pagina WorldZone	46
	5.6	La pagina dei dati macchina	47
6	Con	nclusioni	48

Elenco delle tabelle

2.1	Dati in ingresso al robot	4
2.2	Dati in uscita dal robot	4
2.3	Dati in uscita dal sistema di visione	5
2.4	Dati in ingresso al sistema di visione	5

Elenco dei listati

2.1	OB1	6
2.2	Inputs()	6
2.3	$AlarmsHandler() \dots \dots \dots \dots \dots \dots \dots \dots \dots $	7
2.4	La diagnostica della rete Profinet	8
2.5	HMI()	8
2.6	VisionSystem(): salvataggio programma corrente	0
2.7	thm:polystem	1
2.8	Robot()	3
2.9	RecipesHandler(): ricetta attiva	5
2.10	$\label{eq:RecipesHandler} Recipes Handler (): \ ricetta \ in \ editazione \ . \ . \ . \ . \ . \ . \ . \ . \ . \ $	6
2.11	$\label{eq:RecipeHandler} Recipe Handler (): copiatura ricette \\ \ \ldots \\ \ \ldots \\ \ \ \ldots \\ \ \ \ \ \ \ \ \ \$	6
2.12	$Comunication() \ \dots \ $	6
2.13	Outputs()	9
3.1	CalibData: wobjTable	2
3.2	CalibData: tGripper	2
3.3	CalibData: pHome	3
3.4	CalibData: gli orientamenti	3
3.5	Esempio di utilizzo di Orient ZYX() e EulerZYX() $\ \ldots \ $	3
3.6	World Zone Module: definizione del cubo della area di prelievo $\ \ldots \ \ldots \ \ldots \ 24$	4
3.7	World Zone Module: definizione della World Zone dell'area di prelievo $\dots \dots \dots$	4
3.8	ErrModule	5
3.9	ToolModule: procedura di chiusura della pinza	6
3.10	Tool Module: procedura di apertura della pinza	6
3.11	$\label{thm:prop:signal} VisionSysModule:\ VisionSysCalib \ . \ . \ . \ . \ . \ . \ . \ . \ . \ $	9
3.12	$\label{eq:pick-and-place} Pick And Place Module: Pick \\ \ \dots \\ \ $	2
3.13	$\label{eq:pickAndPlaceModule:Place} PickAndPlaceModule: \ Place \ . \ . \ . \ . \ . \ . \ . \ . \ . \ $	3
3.14	eq:pickAndPlaceModule: SearchProd	4
3.15	$\label{eq:pickAndPlaceModule:PickVisionSysRobT} PickAndPlaceModule: PickVisionSysRobT$	4
3.16	$\label{eq:pickAndPlaceModule:PickAndPlace} PickAndPlace$	5
3.17	PallettizeeModule: Pallettize	5
3.18	$Pallettiza Module: \ Place Prod \ \dots \ \dots \ 36$	6
3.19	PallettizeInterlockingModule: deposito prodotto con incastro 3	7

Elenco delle figure

2.1	La configurazione hardware	3
2.2	Utilizzo del Program_Alarm per la visualizzazione di segnalazioni dinamiche	7
4.1	A sinistra Il software del sistema di visione A destra Il cv-x100	39
5.1	La Home page	41
5.2	La pagina slidin di rimozione prodotti	42
5.3	La pagina del sistema di visione	43
5.4	La pagina delle segnalazioni	44
5.5	La pagina delle ricette	45
5.6	La pagina WorldZone	46
5.7	La pagina dei dati macchina	47

Capitolo 1

Introduzione

1.1 Obiettivi

L'obiettivo del presente lavoro di tesi è quello di realizzare un' applicazione di pallettizzazione per diversi formati di prodotto. I prodotti dovranno essere rilevati dall'area di prelievo da un sistema di visione e successivamente depositati sul pallet. Dal pannello, l'operatore deve avere la possibilità di arrestare il robot in qualsiasi momento, rimuovere uno o più prodotti e riprendere il ciclo dalla prima posizione libera sul pallet. Deve essere inoltre possibile personalizzare il pallet e il tipo di formato del prodotto. I formati in questione sono due: un primo formato standard, completamente bianco, e un secondo formato diviso in due parti, una chiara e una scura. Quest'ultimo dovrà essere posizionato alternando le due parti a formare un incastro.

1.2 Organizzazione dei contenuti

Il presente documento si articola in sei capitoli. Terminata l'iniziale introduzione sul contesto e gli obiettivi che ci si era prefissati, si procede a riportare, nel secondo capitolo, lo sviluppo del software del braccio robotico, soffermandosi sulle peculiarità del linguaggio di programmazione RAPID. Il terzo capitolo è dedicato ad un breve presentazione sul lavoro svolto con il sistema di visione. Nel quarto si tratterà nel dettaglio lo sviluppo del software plc, focalizzandosi sullo scambio dati tra le varie unità. per poi descrivere nel quinto capitolo il risultato dell'interfaccia grafica sviluppata. Per ultimo, il sesto capitolo esporrà le conclusioni ed analizzerà quali potrebbero essere gli eventuali sviluppi futuri.

Capitolo 2

Il software plc

I principi dellautomazione industriale suggeriscono l'utilizzo di sistemi flessibili, capaci di gestire con facilità i vari componenti di un impianto e di permetterne eventuali modifiche durante il funzionamento, riducendo al minimo l'impatto economico e il tempo d'intervento. Questa è la filosofia cardine sulla base della quale è stato ideato e sviluppato il PLC. (*Programmable Logic Controller*).

In questo capitolo, dopo una breve introduzione ai PLC di marca Siemens, verranno illustrati i vari blocchi di programma e le strutture dati che costituiscono software del controllore.

Il PLC utilizzato nell'applicazione oggetto della trattazione è prodotto dalla Siemens AG con il codice S7-1500. La programmazione è stata realizzata mediante il software TIA Portal v.16, fornito dalla casa produttrice, necessario per interfacciarsi con il PLC ed eseguire il caricamento dei programmi in memoria. Numerose altre funzionalità sono disponibili, quali ad esempio la registrazione dei valori assunti da ingressi e uscite, nonché da ogni variabile presente nella memoria del controllore.

2.1 Lo Structured Control Language

Tra i diversi linguaggi disponibili si scelto di utilizzare l'SCL Structured Control Language, scelta motivata dalla maggiore versatilità e velocità di sviluppo rispetto al classico linguaggio a contatti. Le istruzioni SCL utilizzano operatori di programmazione standard, ad es. per l'assegnazione (:=) e le funzioni matematiche (+ per l'addizione, - per la sottrazione, * per la moltiplicazione e / per la divisione). SCL utilizza anche operazioni standard di controllo del programma PASCAL quali IF-THEN-ELSE, CASE, REPEAT-UNTIL, GOTO e RETURN. È possibile utilizzare qualsiasi riferimento PASCAL per gli elementi sintattici del linguaggio di programmazione SCL. [5]

2.2 La configurazione hardware

La configurazione hardware in TIA Portal comprende la configurazione dei dispositivi composta da hardware dei sistemi di automazione, apparecchiature da campo intelligenti e hardware per la visualizzazione. Nella *Vista di rete* è possibile stabilire la comunicazione tra i diversi componenti, i quali vengono prelevati dai cataloghi e inseriti nella configurazione hardware. Per i dispositivi non appartenenti alla famiglia Siemens è prima necessario importare i GSDML, file forniti nel linguaggio GSDML (General Station Description Markup Language) che contengono le proprietà specifiche dell'apparecchiatura di campo.




Figura 2.1: La configurazione hardware

2.3 Il protocollo Profinet

Il protocollo di comunicazione standard per i controllori Siemens è il Profinet, un protocollo di comunicazione basato sulla tecnologia Ethernet. Nasce come evoluzione del Profibus e permette di collegare i dispositivi di campo ai sistemi gestionali dellazienda, consentendo una comunicazione in tempi brevissimi (dellordine del millisecondo). Il Profinet utilizza il protocollo TCP/IP: la comunicazione con i dispositivi di campo avviene tramite lutilizzo di indirizzi IP (Internet Protocol) e di Subnet Mask. Esso permette di collegare tra loro fino a 1024 dispositivi con una velocità di trasmissione di 100 Mbit/s e non ci sono limiti di distanza tra i dispositivi collegati. Queste caratteristiche rendono oggi il Profinet il protocollo più utilizzato nella gestione della comunicazione trai dispositivi di un impianto e il PLC.[1]

2.4 Lo scambio dati

Il ruolo principale del controllore all'interno del sistema è la gestione della comunicazione tra le diverse unità.

Il plc scambia con il robot 124 bytes in ingresso e 124 bytes in uscita . La memoria in ingesso, mappata a partire dall'indirizzo %I1000.0, è salvata nel dato RobotInput di tipo "typeRobotInput" . Allo stesso modo i dati in uscita, dall'indirizzo %Q1000.0, sono scritti sulla variabile RobotInput di tipo "typeRobotOutput".

${\bf RobotInput}$

Varibile	Tipo	Descrizione
commands	typeRobotCommands	robot System Input
interrupts	typeRobotInterrupts	robot interrupts
toolJudgeValue	Array[015] of Bool	vision System Tool Judge
resultReadyFlag	Bool	if TRUE result can be read
commandReadyFlag	Bool	if TRUE cv-x can execute a new command
runMode	Bool	if TRUE cv-x is in run mode
commandCompleteFlag	Bool	if TRUE cv-x completed the command
speedRef	UInt	robot speed set by HMI
removedProd	UInt	removed product index
index	UInt	index
mission	UInt	reserved
productsNum	UInt	products number
productsType	UInt	products type
pickOffs	UInt	pick z offset
placeOffs	UInt	place z offset
palletX	UInt	pallet x coord
palletY	UInt	pallet y coord
palletZ	UInt	pallet z coord
palletR	UInt	pallet rotation (0-45)
coordinates	typeCoordinates	product's coordinates in robot frame
commandResult	DInt	command's result
commandData	Array[12] of DInt	command's result data

Tabella 2.1: Dati in ingresso al robot

${\bf Robot Output}$

Varibile	Tipo	Descrizione
worldzone	typeRobotWorldZone	robot worldzones
motorOnState	Bool	if TRUE motors are in motor On state
emergencyStop	Bool	if TRUE robot is in emergency stop state
executionError	Bool	if TRUE an execution error occurred
isGripperOpen	Bool	if TRUE the gripper is opened
trgControl	Bool	if TRUE robot request a vision system trig
resultAckFlag	Bool	if TRUE robot has received the vision
		system result
isReady	Bool	if TRUE robot and vision system are
, and the second		inizialized
wProdNotFound	Bool	if TRUE there is no product to pick
autoOn	Bool	if TRUE robot is in auto mode
cycleOn	Bool	if TRUE robot program is cycling
notMoving	Bool	if TRUE robot is not moving
limitSpeed	Bool	if TRUE robot speed is limited
commandRequestFlag	Bool	if TRUE robot is sending a vision system command
alarm	Bool	if TRUE a robot error occurred
isGroup1	Bool	if TRUE robot requests a products in range -90 +90
isPlaced	Bool	if TRUE a new product was placed
isStored	Bool	if TRUE a new product was stored
isGripperFull	Bool	if TRUE gripper is full
actProcedure	UInt	current robot procedure
index	UInt	index
errorNumber	DInt	errorNumber
commandNumber	DInt	vision system command number)
commandParameter	Array[12] of DInt	vision system command parameters

Tabella 2.2: Dati in uscita dal robot

Al sistema di visione sono dedicati gli indirizzi a partire da %10.0 e %20.0. Le strutture dati "" e "typeVisionSysOutput" ricalcano lo schema degli ingressi e delle uscite standard del produttore. Le rispettive variabili PLC sono VisionSysInput e VisionSysOutput.

${\bf Visyion System Output}$

Varibile	Tipo	Descrizione
commandCompleteFlag	Bool	if TRUE cv-x command is completed
commandErrorFlag	Bool	if TRUE cv-x error occurred
commandReadyFlag	Bool	if TRUE cv-x can execute a new command
resultReadyFlag	Bool	if TRUE cv-x result can be read
trgReadyStatus	Bool	if TRUE cv-x is ready for a new
orgically Status		trigCommand
trgAckStatus	Bool	if TRUE cv-x has received the trigCom-
organis tartas		mand
busy	Bool	if TRUE cv-x is busy
error	Bool	if TRUE cv-x is in error state
run	Bool	if TRUE cv-x is in run state
toolJudgeValue	Array[063] of Bool	0=NG 1=OK
errorCode	Int	error code
totalCount	DInt	total trig count
commandResult	DInt	command result code
commandData	Array[16] of DInt	command data
resultData	Array[1200] of DInt	tools result data

Tabella 2.3: Dati in uscita dal sistema di visione

${\bf Vision System Input}$

Varibile	Tipo	Descrizione
commandRequestFlag	Bool	if TRUE a command is requested
resultAckFlag	Bool	if TRUE result has been received
errorResetRequestFlag	Bool	if TRUE error reset is requested
trgControl	Bool	if TRUE a new trig is requested
reset	Bool	if TRUE reset is requested
$\operatorname{commandNumber}$	DInt	command number
commandParameter	Array[1510] of DInt	command parameters

Tabella 2.4: Dati in ingresso al sistema di visione

2.5 La struttura del programma

Ogni progetto TIA Portal è costituito da blocchi che vengono richiamati allinterno di un blocco principale e/o di altri blocchi. I blocchi si possono classificare in tre categorie:

- Blocchi Organizzativi (OB): costituiscono l'interfaccia tra il sistema operativo e il programma utente. Alcuni esempi sono l'OB1, che esegue l'elaborazione ciclica del programma, e l'OB100, che descrive il comportamento del sistema di automazione all'avvio.
- Funzioni (FC): blocchi programmabili adatti a realizzare logiche di funzionamento di impianto. Una FC contiene un programma che viene eseguito ogni qualvolta essa viene richiamata da un altro blocco di codice. Per ogni funzione è possibile dichiarare dei parametri di ingresso, di uscita, di ingresso/uscita, nonché delle variabili temporanee. Questultime vengono memorizzate nello stack dei dati locali e vanno perdute dopo lelaborazione dellFC.
- Blocchi Funzionali (FB): blocchi programmabili simili alle FC, ma con la possibilità di definire al loro interno variabili con memoria.

È possibile scrivere l'intero programma utente nel OB1 (programmazione lineare) oppure, quando il sistema risulta complesso ed è necessaria una certa formalità e coerenza nellorganizzazione, si richiamano al suo interno diversi sottoprogrammi(programmazione procedurale). Questultima tecnica à stata seguita per compilare il programma dellapplicazione descritta in questa relazione. Il blocco OB1 viene richiamato dal s.o. ad ogni ciclo (scan). È in questo blocco che andranno messe le chiamate agli altri blocchi del programma utente.

In questa sezione sono illustrati i diversi blocchi che costituiscono l'applicazione.

2.5.1 Il blocco OB Main

Ogni Programma per PLC Siemens deve contenere un blocco organizzativo di sottotipo Main chiamato nello stesso modo. Esso rappresenta la prima parte di codice ad essere eseguita all'avvio della CPU e viene richiamato all'inizio di ogni ciclo di scansione.

nel caos in esame, si è scelto di utilizzare il paradigma della programmazione procedurale, richiamando all'interno dell' OB1 i diversi sottoprogrammi che costituiscono l'applicazione.

```
"Inputs"();
"AlarmsHandler"();
"UnitsHandler"();
"StateHandler"();
"RecipesHandler"();
"Comunication"();
"Outputs"();
```

Listato 2.1: OB1

2.5.2 Il blocco FC Inputs

In questo blocco i dati provenienti dalla periferia sono salvati nelle rispettive variabili globali

```
"RobotData".outputs := "robotOutput";
"VisionSystemData".outputs := "visionSystemOutput";

Listato 2.2: Inputs()
```

2.5.3 Il blocco FC AlarmsHandler

Il blocco AlarmsHandler è adibito alla gestione degli allarmi e delle rispettive segnalazioni.

In presenza di un allarme proveniente dal robot o dal sistema di visione il sistema si deve arrestare e deve essere visualizzata una segnalazione sul pannello operatore.

Tra le diverse modalità di gestione delle segnalazioni possibili in TIA Portal si è scelto di utilizzare l'FB Program_Alarm. La scelta è stata motivata dalla possibilità di visualizzare testi dinamici all'interno della segnalazione, velocizzando la fase di sviluppo e la flessibilità del codice.



Figura 2.2: Utilizzo del Program_Alarm per la visualizzazione di segnalazioni dinamiche

Gli allarmi da ABB

Gli allarmi provenienti da ABB sono segnalati dal fronte di salita del segnale "RobotData". Outputs. Alarm. La variabile visualizzata all'interno della segnalazione è il codice dell'errore contenuto nella variabile "RobotData". Outputs. Error Number. Per gli errori più significativi è stata sviluppata una funzione per convertire l'informazione da numero a stringa per una maggiore chiarezza. Per gli errori standard dei robot ABB si è scelto di rimandare alla documentazione del produttore.

```
CASE #errNo OF
2
      "ERR_OPEN_GRIPPER_TIMEOUT":
3
          #errStrig := 'pinza non aperta entro il timeout';
      "ERR_PROD_NOT_FOUND":
          #errStrig := 'prodotto non rilevato in area di prelievo';
6
      "ERR_RESULT_TIMEOUT":
          #errStrig := 'risultato del sistema di visione non ricevuto entro il timeout';
      "ERR_MOTION_INTERRUPTED":
          #errStrig := 'moto interrotto';
      "ERR_PICK_INTERRUPTED":
          #errStrig := 'moto interrotto durante la fase di prelievo';
      "ERR_PLACE_INTERRUPTED":
13
          #errStrig := 'moto interrotto durante la fase di posizionamento';
14
      "ERR_PRODS_REMOVED":
          #errStrig := 'prodotti rimossi';
16
      "ERR_STORE_INTERRUPTED":
17
          #errStrig := 'moto interrotto durante la fase di deposito';
18
19
      "ERR_COLLISION":
20
          #errStrig := 'collisione';
21
      ELSE
22
          #errStrig := CONCAT(IN1 := DINT_TO_STRING((#errNo)), IN2 := ', leggere la
      documentazione ABB per ulteriori dettagli');
```

```
24 END_CASE;
```

Listato 2.3: AlarmsHandler()

Una segnalazione aggiuntiva informa della mancanza di prodotti in aria di prelievo, la quale causa l'arresto del ciclo.

Gli allarmi da CV-X

Per quanto riguarda gli allarmi del sistema di visione, viene attivata una segnalazione al fronte di salita del bit di errore. Nel testo viene riportato il codice di errore e si rimanda direttamente al manuale Keyence.

La diagnostica della rete Profinet

La diagnostica della rete Profinet è gestita con l'utilizzo del FB Siemens DeviceStates().

, la quale consente di richiamare una determinata informazione di stato concernente tutte le unità in un sistema IO, ad esempio tutti gli IO Device all'interno di un sistema PROFINET IO oppure tutti gli slave DP all'interno di un sistema master DP. Nel listato 2.4 è mostrata la diagnostica degli IO Device/slave DP disturbati della rete "Local PROFINETIO-System. Il risultato è salvato all'interno dell'array di booleani profinetState.

```
profinetState);
#ret := DeviceStates(LADDR := "Local~PROFINET_IO-System", MODE := 2, STATE := #
```

Listato 2.4: La diagnostica della rete Profinet

2.5.4 Il blocco FC UnitsHandler

In questa FC sono richiamati i blocchi per la gestione delle diverse unità:

Il blocco FC HMI

In questo blocco è gestita la visualizzazione dei prodotti del pallet, dipendente dal formato del prodotto, e la rimozione dei dati dal pallet.

```
REGION Pallet Visualization
21
           // if a product is placed, save his value in prods array
22
           REGION countProductPlaced
23
               // count products and find last product
               "HMIData".actProdsCtn := 0;
26
               FOR #i := 1 TO "MAX_PRODUCTS" DO
                   IF "HMIData".prods[#i] THEN
30
31
                        "HMIData".actProdsCtn += 1;
32
                        #lastprod := #i;
34
35
                   END_IF;
36
37
               END_FOR;
           END REGION
40
41
           // find current level
42
           "HMIData".actLevel := "CalcLevel"(prod := #lastprod, dim := 2);
44
           REGION Update Products Placed
45
               //if index is in range, update the corresponding array element
               IF "RobotData".outputs.index >= 1 AND "RobotData".outputs.index <= 20 THEN</pre>
49
50
                   IF "RobotData".outputs.isPlaced THEN
51
                       //if a product is placed, increase array
53
54
                        "HMIData".prods["RobotData".outputs.index] := TRUE;
55
                   ELSIF "RobotData".outputs.isStored THEN
58
                        //if a product is stored, reduce array
59
60
                        "HMIData".prods["RobotData".outputs.index] := FALSE;
                   END_IF;
62
63
               END_IF;
64
           END_REGION
           REGION Remove Products
68
69
           //{
m If} a product is removed by HMI, clear the corresponding array element
               IF "HMIData".robotInputs.interrupts.removeProd THEN
                   //check if index is in range!
72
                   IF "HMIData".robotInputs.removedProd >= 1 AND "HMIData".robotInputs.
73
       removedProd <= "MAX_PRODUCTS" THEN
                   // if removed product isn't in the last level, remove all superior
      products
                       REGION Remove Superior Products
75
```

```
//the start level to remove is the first one over the removed
76
       product level
                            #level := "CalcLevel"(prod := "HMIData".robotInputs.
77
       removedProd, dim := "RobotData".inputs.word3);
                            //ex: level=2 dim=2 => startprod=5
                            #startprod := (#level * "RobotData".inputs.word3) + 1;
                            //check if index is in range!
80
                            IF #startprod >= 1 AND #startprod <= "MAX_PRODUCTS" THEN</pre>
                                 FOR #i := #startprod TO "MAX_PRODUCTS" DO
                                     "HMIData".prods[#i] := FALSE;
84
                                END_FOR;
85
                            END_IF;
                        END_REGION
                        "HMIData".prods["HMIData".robotInputs.removedProd] := FALSE;
                    END_IF;
93
94
                END_IF;
95
           END_REGION
97
98
           CASE "RobotData".inputs.productsType OF
99
100
                0:
                    "HMIData".pallet1 := "HMIData".prods;
                1:
                    "HMIData".pallet2 := "HMIData".prods;
104
           END_CASE;
       END_REGION
107
```

Listato 2.5: HMI()

Il blocco FC VisionSys

In questo blocco è gestita la comunicazione con il sistema di visione.

Ogni formato ha un rispettivo programma all'interno del sistma di visione. Al variare del programma viene salvato il suo valore nella variabile "VisionSystemData".actProg.

```
"VisionSystemData".actProg := "VisionSystemData".outputs.commandData
[2];

END_CASE;

END_IF;

END_IF;

END_REGION
```

Listato 2.6: VisionSystem(): salvataggio programma corrente

In base al valore di "VisionSystemData".actProg vengono letti diversi dati in uscita dal sistema di visione.

```
2 REGION Read Vision System Program Outputs
3 //For each program number, read different data based
4 //on cv-x profinet outputs configuration
      CASE "VisionSystemData".actProg OF
          2: //interlocking
               REGION VisionSystemProgramInterlocking
                   "VisionSystemData".statistics.total := "VisionSystemData".outputs.
10
      resultData[1];
                   \verb"VisionSystemData".statistics.OK := \verb"VisionSystemData".outputs.
      resultData[2];
                   "VisionSystemData".statistics.NG := "VisionSystemData".outputs.
12
      resultData[3]:
13
                   "VisionSystemData".executionTime := "VisionSystemData".outputs.
      resultData[4];
                   "VisionSystemData".prodsCtn := "VisionSystemData".outputs.resultData
16
      [5];
17
                   #resultDataIndex := 6;
18
                   #arraySize := 8; // max number of detected products
19
20
                   //Save coordinates(X,Y,Rz) from vision system data results outputs
21
                   FOR #i := 0 TO #arraySize - 1 DO
                       "VisionSystemData".arrayCoordinates[#i].x := "VisionSystemData".
23
      outputs.resultData[#resultDataIndex];
                       #resultDataIndex := #resultDataIndex + 1;
24
                   END_FOR;
25
                   FOR #i := 0 TO #arraySize - 1 DO
27
                       "VisionSystemData".arrayCoordinates[#i].y := "VisionSystemData".
28
      outputs.resultData[#resultDataIndex];
                       #resultDataIndex := #resultDataIndex + 1;
29
                   END_FOR;
30
31
                   FOR #i := 0 TO #arraySize - 1 DO
32
                       "VisionSystemData".arrayCoordinates[#i].angle := "VisionSystemData
33
      ".outputs.resultData[#resultDataIndex];
                       #resultDataIndex := #resultDataIndex + 1;
34
                   END_FOR;
35
36
               END_REGION
```

```
10: //palletization
               REGION VisionSystemProgramPallettization
                   \verb"VisionSystemData".coordinates.x := \verb"VisionSystemData".outputs.
                   "VisionSystemData".coordinates.y := "VisionSystemData".outputs.
44
      resultData[2];
                   "VisionSystemData".coordinates.angle := "VisionSystemData".outputs.
       resultData[3];
                   "VisionSystemData".statistics.total := "VisionSystemData".outputs.
47
      resultData[4];
                   "VisionSystemData".statistics.OK := "VisionSystemData".outputs.
      resultData[5];
                   "VisionSystemData".statistics.NG := "VisionSystemData".outputs.
49
      resultData[6]:
50
                   "VisionSystemData".executionTime := "VisionSystemData".outputs.
      resultData[7];
52
               END REGION
53
          11.
               REGION VisionSystemProgramCalibration
56
                   "VisionSystemData".statistics.total := "VisionSystemData".outputs.
      resultData[1];
                   \verb|"VisionSystemData".statistics.OK| := \verb|"VisionSystemData".outputs.|
      resultData[2];
                   "VisionSystemData".statistics.NG := "VisionSystemData".outputs.
59
      resultData[3];
60
                   "VisionSystemData".executionTime := "VisionSystemData".outputs.
61
      resultData[4];
62
                   #resultDataIndex := 5;
                   #arraySize := 5; // max number of detected products
                   //Save coordinates(X,Y,Rz) from vision system data results outputs
66
                   FOR #i := 0 TO #arraySize - 1 DO
67
68
                       \verb"VisionSystemData".arrayCoordinates[\#i].x := \verb"VisionSystemData"."
       outputs.resultData[#resultDataIndex];
                       #resultDataIndex := #resultDataIndex + 1;
70
71
                       "VisionSystemData".arrayCoordinates[#i].y := "VisionSystemData".
       outputs.resultData[#resultDataIndex];
                       #resultDataIndex := #resultDataIndex + 1;
74
                   END_FOR;
75
                   "VisionSystemData".prodsCtn := 5;
78
               END_REGION
79
       END_CASE;
82
```

Il blocco FC Robot

In questo blocco è gestita la comunicazione con il robot, dipendente dalla missione in esecuzione. Per il formato standard vengono sempre inviate le coordinate di prelievo del primo prodotto rilevato. Per la pallettizzazione con in castro il sistema di visione invia i dati di tutti i prodotti al plc, il quale seleziona le coordinate del primo prodotto appartenente al gruppo richiesto dal robot. In caso di ricerca fallita, viene inviato al robot un giudizio negativo. Infine per la procedura di calibrazione vengono inviati in sequenza i punti calibrati per definire il sistema di riferimento all'interno del robot.


```
1 IF "StateData".state = "STATE_RUNNING" THEN
      CASE "RobotData".inputs.mission OF
          "PROGRAM_PALLET_INTERLOCKING":
               REGION InterlockingProgram
                   //Build an interlocking pallet.
                   //The products are divided in two groups based on
                   //their angle (90-(-90))
                   //If Vision system detect at least one products,
13
                   //checks all products to find one
14
                   IF "VisionSystemData".outputs.toolJudgeValue[0] THEN
                       REGION "SelectPickCoordinates"
18
19
                           //inizialize loop index and set judge value false.
20
                           //If no product is found, judge value will remain false,
21
                           //otherwise it will become true
22
23
                           #i := 0;
24
                           "VisionSystemData".outputs.toolJudgeValue[0] := false;
25
                           //loop to find the right product to pick
27
                           //loop ends when all products are checked or a right product
28
      is found
29
                           WHILE #i < "VisionSystemData".prodsCtn AND NOT "
30
      VisionSystemData".outputs.toolJudgeValue[0] DO
31
                               //Check if a product belongs to the right group based on
      its angle
33
                               #isGroup1 := NOT ("VisionSystemData".arrayCoordinates[#i].
34
      angle / 1000 > = -90
                                                    AND "VisionSystemData".
35
      arrayCoordinates[#i].angle / 1000 <= 90);
36
                               "VisionSystemData".outputs.toolJudgeValue[0] := ("
      RobotData".outputs.isGroup1 = #isGroup1);
39
```

```
//if a right product is found, save coordinates for robot
40
                                 IF "VisionSystemData".outputs.toolJudgeValue[0] THEN
41
42
                                      \verb"VisionSystemData".coordinates := \verb"VisionSystemData".
43
       arrayCoordinates[#i];
44
                                 END_IF;
45
46
                                 #i := #i + 1;
                             END_WHILE;
49
50
                        END_REGION
51
53
                    END_IF;
54
               END_REGION
           "PROGRAM_CALIBRATION":
               REGION CalibrationProgram
59
60
                    IF "RobotData".outputs.index < "VisionSystemData".prodsCtn AND "</pre>
61
       RobotData".outputs.index >= 0 THEN
62
                        \verb"VisionSystemData".coordinates := \verb"VisionSystemData".
       arrayCoordinates["RobotData".outputs.index];
                        "RobotData".inputs.index := "RobotData".outputs.index;
65
                    END_IF;
66
67
               END_REGION
68
       END_CASE;
72 END_IF;
```

Listato 2.8: Robot()

2.5.5 Il blocco FC StateHandler

In questo blocco è gestita la macchina a stati schematizzato nel grafico sottostante.



2.5.6 Il blocco FC RecipeHandler

Per rendere il sistema personalizzabile dal cliente finale sono state predisposte 20 ricette. Per la loro gestione è stato definita l'UDT "typeRecipe" con i seguenti dati:

- description: Wstring, descrizione della ricetta;
- palletX: Real, coordinata X del centro del pallet;
- palletY: Real, coordinata Y del centro del pallet;
- palletZ: Real, coordinata Z del centro del pallet;
- palletR: Real, rotazione del pallet;
- pickOffs: Uint, offset di approccio in prelievo;
- placeOffs: Uint, offset di approccio in deposito;
- productsNum: Uint, numero dei prodotti da pallettizzare;
- productsType: Uint, formato del prodotto;

Le ricette sono salvate nel controllore all'interno di un blocco dati con memoria a ritenzione e editabili dall'operatore da HMI. Quando la macchina la macchina non si trova in modalità manuale è possibile modificare la ricetta attiva:

Listato 2.9: RecipesHandler(): ricetta attiva

È sempre possibile selezionare e editare le ricette. Quando una ricetta viene selezionata per l'editazione i dati vengono caricati dall'archivio nella struttura "RecipeEdit"; premendo il pulsante a pannello "Save" i dati della ricetta in editazione vengono salavti nell'archivio.

```
1 REGION Edit to Recipes
      IF "Hmi".Recipe.Save THEN
          "Recipes".Recipe["Hmi".Recipe.EditNumber] := "RecipeEdit";
          "Hmi".Recipe.Save := false;
      END_IF;
10 END_REGION
12 REGION Recipes to Edit
      IF "Hmi".Recipe.EditNumber <> "Hmi".Recipe.EditNumberFromHmi THEN
14
          "Hmi".Recipe.EditNumber := "Hmi".Recipe.EditNumberFromHmi;
16
          "RecipeEdit" := "Recipes".Recipe["Hmi".Recipe.EditNumber];
18
      END_IF;
21
22 END_REGION
```

Listato 2.10: RecipesHandler(): ricetta in editazione

L'operatore ha inoltre la possibilità di copiare una ricetta su un'altra. Se la ricette sorgenete e destinazione corrispondo la funzione è disabilitata.

```
REGION Recipe Copy

"Hmi".Recipe.DestinationDescription := "Recipes".Recipe["Hmi".Recipe.

DestinationNumber].description;

"Hmi".Recipe.SourceDescription := "Recipes".Recipe["Hmi".Recipe.SourceNumber].

description;

"Hmi".Recipe.EnableCopy := "Hmi".Recipe.DestinationNumber <> "Hmi".Recipe.

SourceNumber;

IF "Hmi".Recipe.EnableCopy AND "Hmi".Recipe.Copy THEN

"Recipes".Recipe["Hmi".Recipe.DestinationNumber] := "Recipes".Recipe["Hmi".

Recipe.SourceNumber];

"Hmi".Recipe.Copy := false;

END_IF;

END_REGION
```

Listato 2.11: Recipe Handler(): copiatura ricette

2.5.7 Il blocco FC Comunication

In questo blocco vengono cambiati i dati tra le diverse unità

```
1 REGION Write Robot Data
2
3 REGION From Vision System (@\textcolor{black}{to}@) Robot
```

```
//write robot values from Vision System
5
          \verb|"RobotData".inputs.commandCompleteFlag := \verb|"VisionSystemData".outputs.||
      commandCompleteFlag;
          \verb|"RobotData".inputs.commandReadyFlag| := \verb|"VisionSystemData".outputs.|
      commandReadyFlag;
           "RobotData".inputs.runMode := "VisionSystemData".outputs.run;
           "RobotData".inputs.resultReadyFlag := "VisionSystemData".outputs.
12
      resultReadyFlag;
13
          "RobotData".inputs.coordinates := "VisionSystemData".coordinates;
14
          REGION CommandData
16
               "RobotData".inputs.commandData[1] := "VisionSystemData".outputs.
17
      commandData[1];
               "RobotData".inputs.commandData[2] := "VisionSystemData".outputs.
      commandData[2];
          END REGION
19
20
          REGION ToolJudgeValues
21
               "RobotData".inputs.toolJudgeValue[0] := "VisionSystemData".outputs.
      toolJudgeValue[0];
               "RobotData".inputs.toolJudgeValue[1] := "VisionSystemData".outputs.
23
      toolJudgeValue[1];
          END_REGION
26
27
      END_REGION
28
30
          //write robot values from HMI
31
          "RobotData".inputs.commands := "HMIData".robotInputs.commands;
          //value of robot program to run (check robot code for details)
          "RobotData".inputs.mission := "HMIData".robotInputs.mission;
35
36
          REGION interrupts
37
              // write Interrupts
              "RobotData".inputs.interrupts.startMove := "HMIData".robotInputs.
      interrupts.startMove;
               "RobotData".inputs.interrupts.stopMove := "HMIData".robotInputs.interrupts
40
      .stopMove;
               "RobotData".inputs.interrupts.exitCycle := "HMIData".robotInputs.
      interrupts.exitCycle OR "StateData".stopRobot;
               "RobotData".inputs.interrupts.removeProd := "HMIData".robotInputs.
42
      interrupts.removeProd;
              "RobotData".inputs.interrupts.changeSpeed := ("RobotData".inputs.speedRef
      <> "HMIData".robotInputs.speedRef);
          END_REGION
44
45
          "RobotData".inputs.speedRef := "HMIData".robotInputs.speedRef;
          REGION recipeData
               //write number of products and dimension for palletization
49
               "RobotData".inputs.productsNum := "RecipeActive".productsNum;
50
```

```
"RobotData".inputs.productsType := "RecipeActive".productsType;
51
52
               "RobotData".inputs.pickOffs := "RecipeActive".pickOffs;
53
               "RobotData".inputs.placeOffs := "RecipeActive".placeOffs;
54
               "RobotData".inputs.palletX := "RecipeActive".palletX;
               "RobotData".inputs.palletY := "RecipeActive".palletY;
57
               "RobotData".inputs.palletZ := "RecipeActive".palletZ;
               "RobotData".inputs.palletR := "RecipeActive".palletR;
           END_REGION
61
62
           //get number of products removed by HMI
           "RobotData".inputs.removedProd := "HMIData".robotInputs.removedProd;
       END_REGION
66
70 REGION Write Vision System Data
71
       "VisionSystemData".inputs.trgControl := "RobotData".outputs.trgControl OR "HMIData
72
       ".visionSystemInput.trgControl;
       "VisionSystemData".inputs.resultAckFlag := "RobotData".outputs.resultAckFlag OR "
       HMIData".visionSystemInput.resultAckFlag;
       "VisionSystemData".inputs.commandRequestFlag := "RobotData".outputs.
       {\tt commandRequestFlag\ OR\ "HMIData".visionSystemInput.commandRequestFlag;}
       IF "RobotData".outputs.commandRequestFlag THEN
           "VisionSystemData".inputs.commandNumber := "RobotData".outputs.commandNumber;
78
           "VisionSystemData".inputs.commandParameter[1] := "RobotData".outputs.
       commandParameter[1];
           "VisionSystemData".inputs.commandParameter[2] := "RobotData".outputs.
80
       commandParameter[2];
81
       ELSIF "HMIData".visionSystemInput.commandRequestFlag THEN
           "VisionSystemData".inputs.commandNumber := "HMIData".visionSystemInput.
84
       commandNumber;
           "VisionSystemData".inputs.commandParameter[1] := "HMIData".visionSystemInput.
85
       commandParameter[1];
           "VisionSystemData".inputs.commandParameter[2] := "HMIData".visionSystemInput.
       commandParameter[2];
87
       END_IF;
91 END REGION
93 REGION Write HMI Data
           //convert value to ms
95
           "HMIData".executionTime := "VisionSystemData".executionTime / 1000;
           "HMIData".statistics := "VisionSystemData".statistics;
           REGION robotCoordinates
100
               //convert value to mm
```

```
"HMIData".coordinates.x := "VisionSystemData".coordinates.x / 1000;
"HMIData".coordinates.y := "VisionSystemData".coordinates.y / 1000;

//convert value to grades

"HMIData".coordinates.angle := "VisionSystemData".coordinates.angle / 1000;

END_REGION

END_REGION

END_REGION
```

Listato 2.12: Comunication()

2.5.8 Il blocco FC Outputs

In questo blocco i dati globali vengono scritti sulla periferia.

```
"visionSystemInput" := "VisionSystemData".inputs;
"robotInput" := "RobotData".inputs;

Listato 2.13: Outputs()
```

Capitolo 3

Il software del braccio robotico

In questo capitolo sarà descritto il programma del braccio robotico. Per una più chiara comprensione si è deciso di accennare dove necessario alla teoria del linguaggio RAPID, un linguaggio di programmazione di alto livello utilizzato per controllare i robot industriali ABB. Si andranno poi a illustrati i diversi moduli e le routine contenute.

3.1 Introduzione al linguaggio rapid

Il robot a disposizione per l'applicazione oggetto della tesi è IRB1600.6.1,45 con un (Payload) di 6Kg ed estensione del braccio di 1,45m. [2]

Il linguaggio di programmazione standard per questa marca di robot è il Rapid, un linguaggio testuale che assomiglia al Python o a una versione ridotta di un linguaggio C. Sono disponibili tutte le istruzioni principali, come FOR, WHILE, CASE e IF. Sono disponibili tipi di dati standard, array e tipi di dati definiti dall'utente (RECORD). Di seguito una breve introduzione agli elementi di questo linguaggio. Un programma rapid è costituito tipicamente da uno o più moduli, i quali possono comprendere dati e routine diverse. È possibile copiare ciascun modulo, o l'intero programma, su disco, e così via e viceversa. Uno dei moduli contiene la procedura di ingresso, una procedura globale, denominata Main. L'esecuzione del programma indica, in effetti, l'esecuzione di questa procedura.[3]

3.1.1 Le routine

Esistono tre tipologie di routine: procedure, funzioni e trap.

- Una procedura viene utilizzata come un sottoprogramma.
- Una funzione restituisce un valore di un tipo specifico e viene utilizzata come un argomento di un'istruzione.

• Le trap consentono di rispondere agli interrupt. È possibile associare una routine trap a uno specifico interrupt; ad esempio la routine trap viene eseguita automaticamente se si verifica un determinato interrupt impostato precedentemente.

Sono disponibili varie routine standard. Di seguito un breve elenco con alcuni esempi.

- Routine di gestione dei segnali: Set, Reset, SetGO, SetAO, PulseDO.
- Routine di attesa: WaitTime, WaitUntil, WaitAI, WaitAO, WaitDI, WaitDO.
- Routine sui sistemi di riferimento: Offs, RelTool.

3.1.2 I tipi di dati

Per quanto riguarda i tipi di dati, anch'essi si possono dividere in tre categorie:

- Un tipo di dati atomico è tale, poiché che non viene definito in base ad alcun altro tipo e non può essere suddiviso in parti o componenti. Esempi di tipi atomici sono i num e i bool.
- Un tipo di dati record è un tipo composito con componenti nominati ed ordinati, ad esempio pos. Un componente può essere di tipo atomico o record. Un valore record può essere espresso utilizzando una rappresentazione aggregata, ad esempio: valore aggregato record pos [300, 500, depth]. Ad uno specifico componente di dati si può accedere utilizzando il nome di tale componente, ad esempio: pos1.x := 300; attribuzione del componente x di pos1.
- Un tipo di dati alias è, per definizione uguale ad un altro tipo. I tipi alias rendono possibile la classificazione di oggetti di dati.

3.1.3 La istruzioni

Come già accennato in precedenza, il linguaggio RAPID deriva dal linguaggio strutturato, da cui deriva le istruzioni standard quali IF, FOR, WHILE, TEST(corrispondente al CASE).

I linguaggi di programmazione robot si contraddistinguono inoltre per le istruzioni di movimento. Una classica istruzione di movimento in rapid si presenta come:

```
MoveL p1,v500,z10,tool1\wobj:=wobj1;
```

dove:

- MoveL: istruzione di movimento lineare
- p1: robtarget, punto di destinazione
- v500: speeddata, velocità
- z10: zonedata, zona di raccordo
- tool1: tooldata, sistema di riferimento dell' utensile
- wobj1: wobjdata, sistema di riferimento dell'oggetto di lavoro

3.2 Il modulo CalibData

In questo modulo sono definiti i dati necessari alla definizione delle traiettorie. Per i definire i punti di una traiettoria è infatti necessario definire il sistema di riferimento a cui riferiscono i punti e dati del TCO(Tool Center Point), chiamati rispettivamente workobject e tool.

Workobject

Il workobject è un sistema di coordinate utilizzato per descrivere la posizione di un oggetto di lavoro. L'oggetto di lavoro si compone di due sistemi di riferimento: un sistema di riferimento utente e un

sistema di riferimento oggetto. Tutte le posizioni programmate saranno relative al sistema di riferimento oggetto, che è relativo al sistema di riferimento utente, a sua volta relativo al sistema di coordinate universali. Queste ultime corrispondo di default al sistema di riferimento della base così definito:

- L'origine si trova sull'intersezione dell'asse 1 con la superficie di montaggio della base.
- Il piano xy è lo stesso della superficie di montaggio della base.
- L'asse delle x punta in avanti.
- L'asse y punta verso sinistra (dalla prospettiva del robot).
- L'asse z punta verso l'alto.

Nell' applicazione di pallettizzazione sviluppata si è scelto di definire le traiettorie all'interno del sistema di riferimento del tavolo. In questo modo a variazioni dello spostamento del tavolo è sufficiente aggiustare i dati del sistema di riferimento mantenendo invariati i punti di lavoro.

I valori del sistema di riferimento sono i seguenti:

- Il robot non sta sostenendo l'oggetto di lavoro.
- Viene utilizzato il sistema di coordinate utente fisso.
- Il sistema di coordinate utente viene ruotato in angoli di Eulero ZYX di -90r su Z mentre le coordinate della relativa origine sono x= 690, y = -506 e z = 88 mm nel sistema di coordinate universali.
- Il sistema di coordinate oggetto non viene ruotato in angoli di Eulero ZYX di 180ř su X e -90ř su Z mentre le coordinate della relativa origine sono x= 0, y= 200 e z= 30 mm nel sistema di coordinate utente.

```
PERS wobjdata wobjTable := [FALSE,TRUE,"",[[690,-506,88],[0.7071,0,0,-0.7071]],
[[0,0,0],[0,-0.7071,0.7071,0]]];
```

Listato 3.1: CalibData: wobjTable

Tool

Il tooldata viene utilizzato per descrivere le caratteristiche di un utensile, ad esempio una testa di saldatura o una pinza. Queste caratteristiche sono la posizione e l'orientamento del punto centrale dell'utensile (TCP) e le caratteristiche fisiche del carico dell'utensile.

L'utensile per la mia applicazione era una pinza pneumatica con i seguenti valori:

- Il robot sta sostenendo l'utensile.
- Il TCP si trova su un punto a 72,4 mm sulla normale alla flangia di montaggio e a -125 mm lungo l'asse X del sistema di coordinate polso.
- La direzioni dell'utensile sono ruotate in angoli di Eulero ZYX di -180ř su Z e 90ř su Y.
- La massa dell'utensile è di 1,4 kg.
- Il baricentro si trova su un punto a 26,4 mm sulla normale alla flangia di montaggio -40,5 mm lungo l'asse X e 2,8mm sull'asse y del sistema di coordinate del polso,
- Il momento di inerzia è trascurabile.

```
PERS tooldata tGripper:=[TRUE,[[-125,0,72.4],[0,0.7071,0,-0.7071]],

[1.4,[-40.5,2.8,26.4],[1,0,0,0],0.081,0.076,0.019]];

Listato 3.2: CalibData: tGripper
```

Robtarget

Una volta definito il workobject e il tool, è possibile definire i punti della traiettoria. Questi possono essere salvati come posizioni angolari dei singoli assi, jointtarget, oppure come posizione del tcp rispetto ad determinato workobject. In questo caso si parla di robtarget.

Il robtarget pHome nel Listato 3.3 è definito dai seguenti valori:

```
CONST robtarget pHome:=[[-50,400,-200],[1,0,0,0],[-1,1,-1,0],

[9E+09,9E+09,9E+09,9E+09,9E+09,9E+09]];

Listato 3.3: CalibData: pHome
```

- La posizione del robot: x = -50, y = 400, e z = -200 mm nel sistema di coordinate oggetto.
- L'orientamento dell'utensile nella stessa direzione del sistema di coordinate oggetto.
- Il terzo valore indica i dati di configurazione.
- Non sono presenti assi esterni.

L'orientamento

A differenza della maggior parte dei robot industriali, che utilizzano la rappresentazione in angoli di Eulero, l'orientamento nei robot ABB è descritto nella forma di un quaternione costituito da quattro componenti: q1, q2, q3 e q4.

```
CONST orient orVertical:=[1,0,0,0];
CONST orient orHorizontal:=[0.707,0,0,-0.707];
Listato 3.4: CalibData: gli orientamenti
```

In confronto agli angoli di Eulero, i quaternioni presentano funzioni più semplici da comporre ed evitano il problema del blocco cardanico. Inoltre, confrontati con le matrici di rotazione, essi sono più stabili numericamente e forse più efficienti. Lo svantaggio di questa notazione è la perdita di leggibilità e la difficoltà nel calcolarli a mente. Per ovviare a queste problematiche il linguaggio RAPID fornisce due funzioni per la conversione da angoli di Eulero a quaternioni e viceversa, rispettivamente OrienrZYX() e EulerZYX().

```
nRotZ:=EulerZYX(\z,pProd.rot);

IF nRotZ>0 THEN
nRotZ:=nRotZ-90;

ELSE
nRotZ:=nRotZ+90;
ENDIF

pProd.rot:=OrientZYX(nRotZ,0,0);

Listato 3.5: Esempio di utilizzo di OrientZYX() e EulerZYX()
```

Per una maggiore leggibilità sono state definite due costanti per rappresentare l'orientamento, vedi il Listato 3.4, utilizzato per tutto il codice.

3.3 Il modulo WorldZoneModule

All'interno di un' applicazione RAPID è possibile definire delle World Zone, cioè aree di lavoro supervisionate durante i movimenti del robot, sia durante l'esecuzione del programma sia durante il movimento manuale.

Nel modulo WorldZoneModule sono definite le seguenti aree cubiche:

- wzHome: area di Home.
- wzPicking: area di prelievo sottostante la camera del sistema di visione.
- wzPlacing: area di deposito.
- wzWarehouse: area del magazzino.
- wzDangerouse: area al di fuori della zona di lavoro.

Il primo passo per creare una World Zone cubica è definire il volume del cubo con l'istruzione WZBoxDef. L'istruzione successiva WZDOSet viene utilizzata per definire lazione e attivare una World Zone per la supervisione dei movimenti del robot. Dopo avere eseguito questa istruzione, quando il TCP del robot o il robot e gli assi esterni (zona nei giunti) si trovano all'interno della World Zone definita o la stanno per raggiungere, un segnale di output viene impostato sul valore specificato.

```
CONST pos psPickingCorner1:=[-150,200,0];

CONST pos psPickingCorner2:=[-700,800,-800];

VAR wztemporary wzPicking;

VAR shapedata pickingVolume;

WZBoxDef\Inside,pickingVolume,posTableToWCS(psPickingCorner1),

posTableToWCS(pspickingCorner2);

WZDOSet\Temp,wzPicking\Inside,pickingVolume,dowzPicking,1;
```

Listato 3.6: WorldZoneModule: definizione del cubo della area di prelievo

I dati di posizione psPickingCorner1 e psPickingCorner2 descrivono gli angoli opposti del cubo all'interno del sistema di riferimento del tavolo. Dato che l'istruzione WZBoxDef necessita di dati di posizione all'interno del sistema di riferimento mondo, è stato utilizzata la funzione posTableToWCS() mostrata nel Listato 3.7, che utilizza le funzioni standard poseVect() e poseMult() per convertire una posizione nel sistema di riferimento del tavolo in una posizione nel sistema di riferimento mondo.

```
FUNC pos posTableToWCS(pos ps)

RETURN poseVect(poseMult(wobjtable.uframe,wobjtable.oframe),ps);

ERROR
RAISE;

ENDFUNC
```

Listato 3.7: WorldZoneModule: definizione della World Zone dell'area di prelievo

L"utilizzo delle aree di lavoro verrà approfondite in seguito lungo la trattazione.

3.4 Il modulo ErrModule

Molte condizioni di errore che si verificano durante l'esecuzione di un programma possono essere gestite al suo interno senza interromperne l'esecuzione. Questi tipi errori sono rilevati dal sistema, ad esempio la divisione per zero, o dal programma, ad esempio gli errori generati quando viene letto un valore non corretto da un lettore di codice a barre. Se nella routine corrente non è disponibile un gestore errori, viene attivato direttamente il gestore interno del robot, che genera un messaggio di errore e termina l'esecuzione del programma.[4]

Ogni errore è identificato da un numero univoco, l' ERRNO, che può essere utilizzato all'interno del programma per determinare il tipo di errore verificatosi. Inoltre, è possibile utilizzare l'istruzione RAISE per creare un errore nel programma e, successivamente, per chiamare un gestore errori della routine. Con questa istruzione è inoltre possibile propagare un errore da un gestore errori, a partire da una catena di chiamate nidificate fino a un livello superiore, in un'unica operazione.

Nel modulo err Module sono state definite le costanti per gli errori richiamati all'interno delle routine del programma.

```
2 MODULE ErrModule
      CONST errnum errOpenGripperTimeout:=1;
      CONST errnum errProdNotFound:=2;
      CONST errnum errResulTimeOut:=3;
      CONST errnum errMotionInterrupted:=4:
      CONST errnum errPickInterrupted:=5;
      CONST errnum errPlaceInterrupted:=6;
      CONST errnum errProdsRemoved:=7;
      CONST errnum errStoreInterrupted:=9;
13
      PROC errHandler()
14
           ResetOutputs;
16
           setAlarm ERRNO;
17
           ExitCycle;
19
20
      ENDPROC
21
      PROC setAlarm(num nErr)
25
           SetGO goErrorNumber,nErr;
26
           PulseD0\High\PLength:=2,doAlarm;
      ENDPROC
29
30
31 ENDMODULE
```

Listato 3.8: ErrModule

Ogni procedura presenta sempre al suo interno un istruzione RAISE per propagare l'errore al livello successivo. Se l'errore raggiunge il Main() senza essere stato precedentemente gestito, viene richiamata la procedura errHandler():

- 1. vengono resettati gli outputs correnti;
- 2. viene inviato il codice di errore contenuto della variabile ERRNO;
- 3. viene inviato il segnale di allarme;
- 4. l'esecuzione del ciclo termina con l'istruzione ExitCycle.

3.5 Il modulo ToolModule

In questo modulo sono gestire le procedure di apertura e chiusura della pinza. Questa è collegata alla scheda di comunicazione interna al robot, la quale controllo le elettrovalvole. Con i segnali i_GripperExitClose e o_RetractorEntry è quindi possibile comandare e leggere lo stato della pinza.

```
! i_GripperExitClose : 1=closed 0=opened !!o_RetractorEntry : 1=closing 0=opening
      PROC CloseGripper()
2
           IDisable;
           SetDO\Sync,o_RetractorEntry,high;
           WaitTime 0.2;
           Reset doGripperOpen;
           IEnable;
      ERROR
12
           RAISE;
14
      ENDPROC
                          Listato 3.9: ToolModule: procedura di chiusura della pinza
      PROC OpenGripper()
           VAR bool bTimeOut:=FALSE;
           IDisable;
           SetDO\Sync,o_RetractorEntry,low;
           WaitDI i_GripExitClose,low\MaxTime:=2\TimeFlag:=bTimeOut;
           IF bTimeOut THEN
11
12
               RAISE errOpenGripperTimeout;
           ELSE
               Set doGripperOpen;
           ENDIF
15
16
           IEnable;
17
18
      ERROR
           RAISE;
20
21
      ENDPROC
```

Listato 3.10: ToolModule: procedura di apertura della pinza

3.6 Il modulo VisionSysMoule

Per gestire la comunicazione con il sistema di visione è stata sviluppata una libreria di routine. Di seguito una veloce presentazione delle stesse.

La procedura VisionSysResetTrig

Resetta il segnale di trigger per una successiva elaborazione.

```
PROC VisionSysResetTrig()

Set doResultACK;
WaitDI diResultReady,low\MaxTime:=5;
Reset doResultACK;
```

```
7 ERROR
8 RAISE;
9
10 ENDPROC
```

La procedura VisionSysTrig

Invia il segnale di trigger per una nuova elaborazione.

```
PROC VisionSysTrig(\num nDelay)

IF diResultReady=1 VisionSysResetTrig;

IF Present(nDelay) WaitTime nDelay;

PulseDO\high,doTrgControl;

ERROR
RAISE;
ENDPROC
```

La procedura VisionSysSendCommand

Invia un comando al sistema di visione e ritorna i dati del risultato.

```
PROC VisionSysSendCommand(num nCommand,\dnum nParameter1,\dnum nParameter2)
           VAR dnum nCommandParameter1:=0;
           VAR dnum nCommandParameter2:=0;
           IF Present(nParameter1) nCommandParameter1:=nParameter1;
           IF Present(nParameter2) nCommandParameter2:=nParameter2;
           IF diCommandReadyFlag=0 THEN
               Reset doCommandRequestFlag;
10
               WaitDI diCommandReadyFlag, high;
           ENDIF
           SetGO goCommandNumber,nCommand;
           SetGO goCommandParameter1,nCommandParameter1;
15
           SetGO goCommandParameter2,nCommandParameter2;
16
17
           Set doCommandRequestFlag;
           WaitDI diCommandCompleteFlag , high \ MaxTime := 30;
19
           Reset doCommandRequestFlag;
20
21
      ERROR
22
           RAISE ;
24
      ENDPROC
```

La funzione VisionSysCProgram

Restituisce il programma attualmente presente nel sistema di visione.

```
1 FUNC dnum VisionSysCProgram()
2
3 VAR dnum CProgram;
```

```
VisionSysSendCommand READ_PROG_COMMAND;
Cprogram:=GInputDnum(giCommandData2);

RETURN CProgram;

ERROR
RAISE;

ENDFUNC
```

LA procedura VisionSysSetProgram

Se non lo è già, imposta il sistema di visione in modalità RUN. Imposta un nuovo programma all'interno del sistema di visione. Se il programma richiesto è attualmente presente, ne resetta le statistiche.

```
PROC VisionSysSetProgram(dnum nProgram)
           Reset doTrgControl;
           Reset doResultACK;
           IF diRunMode=0 THEN
               VisionSysSendCommand RUN_MODE_COMMAND;
           ENDIF
           IF VisionSysCProgram()<>nProgram THEN
               Set doChangingVisionSysProg;
               VisionSysSendCommand CHANGE_PROG_COMMAND,
               \nParameter1:=1,\nParameter2:=nProgram;
               Reset doChangingVisionSysProg;
14
           ELSE
               VisionSysSendCommand RESET_COMMAND;
16
           ENDIF
17
           VisionSysTrig;
19
20
      ERROR
21
          RAISE ;
22
       ENDPROC
```

La funzione VisionSysRobT

Elabora le coordinate provenienti dal sistema di visione(aiPosX, aiPosY, aiRotZ) in un punto del sistema di riferimento del tavolo. PeVisionSysToTable contiene i dati della trasformata per passare dal sistema di riferimento del sistema si dvisione a quello del tavolo. La funzione poseMult restituisce il risultato della trasformazione.

```
RAISE;
12 ENDFUNC
```

La funzione VisionSysCalib

Per convertire i punti dal sistema di coordinate della camera al sistema di coordinate del tavolo è necessario prima definire le trasformazioni tra i due sistemi. Per questa ragione è stata sviluppata una procedura di calibrazione che utilizza l' istruzione RAPID DefAccFrame, con la quale è possibile definire un sistema di riferimento dopo aver acquisito una serie di destinazioni in due posizioni diverse. Per definire un sistema di riferimento sono sufficienti tre destinazioni, ma per una maggiore accuratezza è necessario utilizzare un array di 5 punti. Dopo aver impostate il corrispettivo programma nel sistema di visione, viene scatta una foto all'area di prelievo e inviati i dati dei punti fissi di calibrazione precedentemente salvati nell'array pCam. Le ccordinate dei punti sono quindi convertite nel sistema di riferimento mondo e utilizzate per definire il sistema di riferimento del sistema di visione.

```
2
      CONST robtarget pCam{5}:=[pCam1,pCam2,pCam3,pCam4,pCam5];
      PERS robtarget pCamVS{5}:=[[[66.7898,168.901,0],[1,0,0,0],[1,0,0,0],[9E+09,9E+09,9
      E+09,9E+09,9E+09,9E+09]],
                                       [[233.956,149.256,0],[1,0,0,0],[1,0,0,0],[9E+09,9E
      +09,9E+09,9E+09,9E+09,9E+09]],
                                       [[147.652,91.9167,0],[1,0,0,0],[1,0,0,0],[9E+09,9E
      +09,9E+09,9E+09,9E+09,9E+09]],
                                       [[53.3243,55.8337,0],[1,0,0,0],[1,0,0,0],[9E+09,9E
      +09,9E+09,9E+09,9E+09,9E+09]],
                                       [[226.856,39.3594,0],[1,0,0,0],[1,0,0,0],[9E+09,9E
      +09,9E+09,9E+09,9E+09,9E+09]]];
      PERS robtarget pCamWCS{5}:=[[[1114,-79,93],[1,0,0,0],[-1,-1,2,1],[9E+09,9E+09,9E
      +09,9E+09,9E+09,9E+09]],
                                       [[1079,84,93],[1,0,0,0],[0,0,1,1],[9E+09,9E+09,9E
      +09,9E+09,9E+09,9E+09]],
                                       [[1033,-8,93],[1,0,0,0],[-1,-1,2,1],[9E+09,9E+09,9
      E+09,9E+09,9E+09,9E+09]],
                                       [[1005,-104,93],[1,0,0,0],[-1,-1,2,1],[9E+09,9E
13
      +09,9E+09,9E+09,9E+09,9E+09]],
                                       [[972,65,93],[1,0,0,0],[0,0,1,1],[9E+09,9E+09,9E
      +09,9E+09,9E+09,9E+09]]];
16
      PERS pos psVisionSysToTable:=[-346.134,419.617,0];
18
19
      PROC VisionSysCalib()
20
          CONST num nCalibPoints:=5;
23
          VisionSysSetProgram N_CALIBRATION_PROG;
24
          SearchProd:
25
26
          FOR i FROM 1 TO nCalibPoints DO
              SetGO goIndex,i-1;
28
29
               ! get point in Word coordinate system
30
              MoveL pCam{i}, vfast, fine, tGripper, \WObj:=wobjTable;
              pCamWCS{i}:=CRobT(\Tool:=tgripper\WObj:=wobj0);
```

```
pCamWCS{i}.rot:=[1,0,0,0];
33
34
                ! Get point in vision system Coordinate system
35
                waituntil GOutput(goIndex)=giIndex;
36
                p \texttt{CamVS\{i\}:=[[(aiPosx/1000)*nresolution,(aiPosy/1000)*nresolution]} \\
       ,0],[1,0,0,0],[1,0,0,0],[9E+09,9E+09,9E+09,9E+09,9E+09,9E+09]];
38
39
           peVisionSys:=DefAccFrame(pCamVS,pCamWCS,nCalibPoints,nmaxErr,nmeanErr);
           mvHome;
42
43
       ERROR
44
           RAISE ;
46
47
       ENDPROC
```

Listato 3.11: VisionSysModule: VisionSysCalib

3.7 Il modulo InterruptsModule

Per comunicare in maniera asincrona rispetto all'esecuzione del programma del robot, è presente un modulo contenente interrupts e trap.

${\bf L'interrupt\ int Stop Move}$

Interrompe il movimento del moto.

```
ISignalDI diIntStopMove, high, intStopMove;

TRAP trapstopmove

StopMove;

Set doNotMoving;

nRemovedProds:=0;

ENDTRAP
```

L'interrupt intStartMove

Riprende il movimento interrotto. Se il robot si trova all'interno del sistema di visione durante la fase di prelievo, si sposta al di fuori ed esegue un nuovo PickVisionSysRobT.

```
ISignalDI diIntStartMove, high, intStartMove;

TRAP trapStartmove

VAR bool bPickRestarted:=FALSE;

Reset doNotMoving;

IF bEnablePickRestart bPickRestarted:=CheckRestartPick();

IF CheckRemovedProds() OR bPickRestarted THEN

RAISE errMotionInterrupted;

ELSE
```

```
StartMove;
          ENDIF
16
17
      ERROR
18
         IF ERRNO = errMotionInterrupted THEN
              ClearPath;
20
               StartMove;
21
         ENDIF
22
           RAISE ;
25
      ENDTRAP
```

L'interrupt intExitCycle

Conclude l'esecuzione del programma.

```
ISignalDI diIntExitCycle, high, intExitCycle;

TRAP trapExitCycle
ResetOutputs;
suctionDisable;
ExitCycle;
ENDTRAP
```

L'interrupt intChangeSpeed

Cambia la velocità del robot.

```
ISignalDI diIntChangeSpeed, high, intChangeSpeed;

TRAP trapChangeSpeed

vfast.v_tcp:=giSpeedRef;

ENDTRAP
```

L'interrupt intRemoveProd

Segnala al robot che un prodotto è stato rimosso.

```
ISignalDI diIntRemoveProd, high, intRemoveProd;

TRAP trapRemoveProds

!if removed Prod isn't in the last level,
!remove all superior products.
!ex: level=2 dim=2 => index=5

pallet.index:=Calclevel(giRemovedProd)*pallet.dim+1;
```

```
12
           WHILE pallet.index<=pallet.nProds DO</pre>
               IF prods{pallet.index}.placed THEN
14
                   prods{pallet.index}.placed:=FALSE;
                    incr nRemovedProds;
17
               incr pallet.index;
18
           ENDWHILE
19
           prods{giRemovedProd}.placed:=FALSE;
21
           Incr nRemovedProds;
22
23
       ENDTRAP
24
```

L'interrupt intTrig

Esegue un VisionSysTrig per il prelievo successivo non appena il robot è al di fuori dell'area visualizzata dal sistema di visione.

```
ISignalDO dowzPicking,low,intTrig;

TRAP trapTrig
PulseDO\High,doTrgControl;
ENDTRAP
7
```

L'interrupt intColl

Riconosce le collisioni ed invia una segnalazione.

```
1
2     IError MOTION_ERR\ErrorId:=204,TYPE_ALL,intErrColl;
3
4     TRAP trapErrColl
5     setAlarm 204;
6     ExitCycle;
7     ENDTRAP
```

3.8 Il modulo PickAndPlaceModule

Questo modulo contiene le diverse procedure sviluppate per eseguire un ciclo di "pick and place".

La procedura Pick

Procedura di prelievo.

```
PROC Pick(robtarget p,num nZOffset)

OpenGripper;
mvTopDown p,nZOffset;
CloseGripper;
Set doGripperFull;
```

```
MoveL Offs(p,0,0,nZOffset),vFast,z50,tGripper\WObj:=wobjTable;
8
9
       ERROR
10
11
           IF ERRNO = errMotionInterrupted THEN
12
               IF nRemovedProds > 0 THEN
13
                    RAISE errProdsRemoved;
14
               ELSE
                    RAISE errPickInterrupted;
               ENDIF
17
           ELSE
18
               RAISE ;
19
           ENDIF
21
       ENDPROC
22
```

Listato 3.12: PickAndPlaceModule: Pick

La procedura Place

Procedura di deposito.

```
PROC Place(robtarget p,num nZOffset)
           mvTopDown p,nZOffset;
           OpenGripper;
           Reset doGripperFull;
           MoveL Offs(p,0,0,nZOffset),vFast,z50,tGripper\WObj:=wobjTable;
      ERROR
12
13
14
           IF ERRNO = errMotionInterrupted THEN
15
               IF NOT gripperFull() Pick p,0;
16
               IF nRemovedProds > 0 THEN
17
                   RAISE errProdsRemoved;
               ELSE
                   RAISE errPlaceInterrupted;
20
               ENDIF
21
22
           ELSE
23
               RAISE ;
           ENDIF
25
      ENDPROC
26
```

Listato 3.13: PickAndPlaceModule: Place

La procedura SearchProd

Procedura di ricerca di un prodotto nella zona di prelievo. Invia una richiesta di elaborazione e attende il risultato. Se l'elaborazione fallisce o non arriva entro 10s, ritenta una una elaborazione. Dopo 3 elaborazioni ritorna un errore.

```
PROC SearchProd()
2
3
           VAR bool bProdFound:=false;
           VAR bool bTimeOut:=false;
           VAR num nFailsCounter:=0;
           WHILE ((NOT bProdFound) AND (nFailsCounter <3)) DO
               WaitDI diResultReady , high \ MaxTime := 10 \ TimeFlag := bTimeOut;
               IF bTimeOut THEN
                   RAISE errResulTimeOut;
               ELSE
                   IF diToolJudge0=1 THEN
15
                        VisionSysResetTrig;
16
                        bProdFound:=true;
                   ELSE
                        Incr nFailsCounter;
                        VisionSysTrig;
20
                   ENDIF
21
               ENDIF
22
           ENDWHILE
24
25
           IF NOT bProdFound RAISE errProdNotFound;
26
       ERROR
           RAISE ;
29
       ENDPROC
```

Listato 3.14: PickAndPlaceModule: SearchProd

La procedura PickVisionSysRobT

Procedura di prelievo con l'utilizzo del sistema di visione. Se la procedura SearchProd si conclude con successo, viene prelevato il prodotto alle coordinate ricevute dal sistema di visione. La rotazione viene mantenuta all'interno di un range di 180

```
PROC PickVisionSysRobT(num nZOffset)
           VAR robtarget pProd;
           VAR num nRotZ;
           SearchProd;
           pProd:=VisionSysRobT();
           nRotZ:=EulerZYX(\z,pProd.rot);
11
           IF nRotZ>0 THEN
12
               nRotZ := nRotZ - 90;
13
           ELSE
14
               nRotZ := nRotZ + 90;
           ENDIF
16
17
           pProd.rot:=OrientZYX(nRotZ,0,0);
18
           Pick pProd,nZOffset;
```

```
21
       ERROR
22
            IF ERRNO = errProdNotFound THEN
23
                ProdNotFoundMessage;
24
                RETRY;
            ELSE
26
                RAISE ;
27
            ENDIF
30
       ENDPROC
31
```

Listato 3.15: PickAndPlaceModule: PickVisionSysRobT

La procedura PickAndPlace

Procedura completa di prelievo e deposito con i dati provenienti dal sistema di visione.

```
PROC PickAndPlace(robtarget pPlace,num nZOffset)
           bEnablePickRestart:=TRUE;
           PickVisionSysRobT nZOffset;
           IWatch intTrig;
           Place pPlace, nZOffset;
11
           bEnablePickRestart:=FALSE;
           ISleep intTrig;
13
14
      ERROR
15
16
           IF ERRNO = errPickInterrupted THEN
               RETRY;
           ELSE
               RAISE ;
20
           ENDIF
21
      ENDPROC
23
```

Listato 3.16: PickAndPlaceModule: PickAndPlace

3.9 Il modulo PallettizeModule

In questo modulo è sviluppata la pallettizzazione vero e propria, utilizzando le routine e dati presentate precedentemente.

Per prima cosa viene impostato il programma all'interno del sistema di visione corrispondente al formato. Successivamente vengono calcolati i dati del pallet con i dati a ricetta selezionati dall'operatore. Il ciclo di prelievo e deposito continua fino al completamento del pallet, dopodiché il ciclo termina con il robot in posizione di Home.

```
PROC Palletize(robtarget p0)

VisionSysSetProgram N_PALLETTIZE_PROG;
```

```
CalcPalletPoints p0;

WHILE PlaceProd() D0
ENDWHILE

mvSafeZ tGripper;
mvHome;

ERROR
RAISE;

ENDPROC
```

Listato 3.17: PallettizeeModule: Pallettize

La funzione PlaceProd

Di seguito la funzione PlaceProd, la quale ritorna TRUE se è stato depositato un nuovo prodotto, FALSE se il ogni elemento dell'array del pallet risulta completo.

```
FUNC bool PlaceProd()
           VAR bool newprodplaced:=FALSE;
           pallet.index:=1;
           WHILE pallet.index <= pallet.nProds AND (NOT newprodplaced) DO
               IF NOT prods{pallet.index}.placed THEN
                   IF gripperFull() THEN
11
12
                        Place prods{pallet.index}.point,nZOffset;
13
                        PickAndPlace prods{pallet.index}.point,nZOffset;
14
                   ENDIF
                   IncrProds;
                   newprodplaced:=TRUE;
18
                   Incr pallet.index;
19
               ENDIF
21
           ENDWHILE
22
23
           RETURN newprodplaced;
24
      ERROR
27
           IF ERRNO=errProdsRemoved THEN
28
               RETURN TRUE;
29
           ELSE
               RAISE ;
31
           ENDIF
32
33
      ENDFUNC
34
```

Listato 3.18: PallettizaModule: PlaceProd

3.10 Il modulo PallettizeInterlockingModule

La presente funzione è del tutto analoga a quella precedente, con la differenza di effettuare una pallettizzazione "con incastro": i prodotti di questo formato, divisi in una parte bianca e una nera, che si devono alternare tra loro. Per questo motivo si è mantenuta invariata la struttura principale, sostituendo la procedura PlaceProduct con la procedura Placeinetrlocking. In base all'indice del prodotto viene richiesto un prodotto appartenente ad un diverso gruppo, identificato dalla rotazione. In caso di ineterruzione del ciclo e successiva rimozione di un prodotto, viene controllato lo stato della pinza: se contiene un prodotto non appartenente al gruppo richiesto, viene scartato e il ciclo riprende da un nuovo prelievo.

La funzione PlaceInterlocking

```
FUNC bool PlaceInterlocking()
           VAR bool newprodplaced:=FALSE;
           pallet.index:=1;
           WHILE pallet.index <= pallet.nProds AND (NOT newprodplaced) DO
               IF NOT Prods{pallet.index}.placed THEN
                    IF gripperFull() THEN
11
                        IF selectGroup() = DOutput(doGroup1) THEN
                            Place Prods{pallet.index}.point,nZOffset;
                            IncrProds;
                        ELSE
                            Store:
                        ENDIF
17
18
                    ELSE
19
                        AlternateProds;
                        PickAndPlace Prods{pallet.index}.point,nZOffset;
20
                        IncrProds;
21
                    ENDIF
                    newprodplaced:=TRUE;
                    Incr pallet.index;
25
               ENDIF
26
27
           ENDWHILE
           RETURN newprodplaced;
30
      ERROR
33
           IF ERRNO = errProdsRemoved THEN
34
               RETURN TRUE;
35
           ELSE
36
               RAISE ;
           ENDIF
38
39
      ENDFUNC
40
```

Listato 3.19: PallettizeInterlockingModule: deposito prodotto con incastro

Capitolo 4

Il sistema di visione

Per visione artificiale si intende linsieme dei processi che mirano a creare un modello approssimato del mondo reale partendo da immagini bidimensionali. I sistemi di visione attualmente in commercio per applicazioni industriali sono strumenti di utilizzo estremamente semplice ed intuitivo. In questo capitolo si andranno ad illustrare i passaggi per lo sviluppo del software per l'applicazione oggetto della trattazione.

Il sistema di visione in dotazione appartiene alla famiglia cv-x della Keyence. Il suo editor di sviluppo è costituito da un un interfaccia grafica semplice ed intuitiva. Ogni programma è costituito da un set di strumenti disponibili dal catalogo. Ogni strumento può essere personalizzato con diversi filtri per migliorare la rilevazione e parametri quali la condizione di giudizio e la percentuale di somiglianza con il modello. Questi strumenti possono essere eseguiti in sequenza, in base a determinate condizioni o in parallelo in funzione dell'applicazione. Le informazioni così ottenute possono poi essere inviate direttamente al robot o, come nel mio caso, al PLC. Nella impostazioni, sotto la voce *Profinet* è quindi possibile definire quali dati scambiare in ingresso e in uscita, sotto forma di bit o word.

4.1 Il software per il formato standard

Nello sviluppo del software per il prodotto monocolore si è proceduto nelle seguenti fasi

- 1. Selezionare lo strumento Valuta con Ricerca Profilo
- 2. Scattare una foto dei prodotti
- 3. Definire il contorno di uno di essi
- 4. Definire la condizione di giudizio: OK se almeno un prodotto visualizzato, NG in caso contrario
- 5. Selezionare la percentuale di somiglianza con il modello: 40% è risultato essere il giusto trade-off. Infine è stata definita la configurazione delle uscite da inviare via Profinet al PLC.



Figura 4.1: A sinistra Il software del sistema di visione A destra Il cv-x100

4.2 Il software per il formato bicolore

Nella pallettizzazione con incastro era necessario riconoscere la rotazione del prodotto in area di prelievo. Per raggiungere questo obiettivo è stato necessario aggirare la limitazione del sistema di visione in dotazione, non in grado di riconoscere i colori. La soluzione trovata è stata differenziare le due parti del prodotto con una piccola irregolarità di forma nel profilo della parte scura. In questo il sistema di visione è stato in grado di riconoscere la differenza di rotazione dei prodotti rispetto al prodotto utilizzato come modello e inviarle al controllore. Quest'ultimo seleziona tra i dati ricevuti le coordinate del prodotto prodotto appartenente al gruppo richiesto dal robot. In caso di applicazioni in cui non è possibile alterarne le caratteristiche si sarebbe potuto aggirare il problema con l'utilizzo di un sistema di visione più performante o con l'applicazione di codici a barre rimovibili in una fase successiva.

4.3 Luminosità e calibrazione

Nell'utilizzo dei sistemi di visioni ci si scontra con diverse problematiche. La prima è quella della luminosità: essi infatti sono estremamente soggetti alle variazioni luminose. per questa ragione per l'utilizzo in un magazzino con una luminosità non controllata è necessario sempre regolare la sensibilità prima di ogni utilizzo.

La seconda difficoltà è quella della calibrazione: i sistemi di visione necessitano di essere perfettamente calibrati per restituire coordinate corrette. Per questa ragione è stato sviluppata una procedura di calibrazione automatica da lanciare dopo ogni modifica della posizione della camera o dopo lunghi periodi di non utilizzo.

Capitolo 5

L'interfaccia uomo-macchina

I sistemi di controllo industriali richiedono molto spesso la possibilità di variare in opera alcune funzioni del sistema controllato, il monitoraggio delle operazioni da parte degli operatori o ancora la possibilità di inviare comandi al sistema in modo rapido e affidabile. A tale scopo vengono utilizzate le interfacce uomo-macchina HMI (Human Machine Interface), che permettono alluomo di operare e interagire direttamente con le macchine mediante una dashboard opportunamente progettata e programmata. In questo capitolo sarà illustrata l'interfaccia uomo-macchina sviluppata per questa applicazione.

5.1 La pagina Home

L'interfaccia è basata su un template comune a tutte le pagine composto dalla dalla barra di stato nella parte alta della pagina e una barra bassa di navigazione. Dopo lavvio dellimpianto, il display visualizza il menu principale, denominato HOME. Questo menu è sempre richiamabile da qualsiasi altra pagina mediante la pressione dellapposito tasto disponibile in basso a sinistra.

Nella barra di stato è mostrato il nome della pagina(1), l'utente attualmente connesso(2), la ricetta attiva(3), lo stato del sistema(4), la presenza di allarmi eventuali allarmi o segnalazioni(5), il feedback di comunicazione con il plc(6) e il logo dell'azienda(7). La barra di navigazione(12) permette di raggiungere le varie pagine dell'interfaccia, da sinistra verso destra.

- 1. La pagina Home
- 2. La pagina di impostazioni
- 3. La pagina di manutenzione
- 4. La pagina delle segnalazioni




Figura 5.1: La Home page

- 5. La pagina delle ricette
- 6. La pagina WorldZone
- 7. La pagina dei dati macchina

Nella zona centrale sulla sinistra(8) sono disposti i comandi per il robot. Dall'alto verso il basso:

- 1. Accensione motori
- 2. Avvia/termina ciclo
- 3. Arresta/riprendi ciclo
- 4. Limitazione velocità
- 5. Arresto di emergenza

Sulla destra(10) è visibile lo stato della pinza o l'eventuale presenza di un prodotto. La parte centrale della pagina è riservata alla visualizzazione dello stato del pallet in costruzione(9): Numero di prodotti attualmente posizionati e il livello raggiunto. Quando il robot non è in movimento è possibile rimuovere un prodotto. L'operatore può selezionare direttamente il prodotto sul pallet oppure, premendo il pulsante(11) alla destra del pallet, aprire la pagina slidin di selezione in Figura 5.2. Una volta selezionato un prodotto si apre un popup di conferma. Una volta ripreso il ciclo il robot deposita sulla prima posizione libera.




Figura 5.2: La pagina slidin di rimozione prodotti

5.2 La pagina di manutenzione

Dalla pagina di manutenzione si accede alla pagina di diagnostica e a quella di gestione del sistema di visione. In Figura 5.3 sono visibili sulla sinistra i dati riguardanti le statistiche delle elaborazione, il tempo di elaborazione e il codice del programma attualmente presente all'interno del sistema(1). Sulla destra sono presenti i comandi per forzare i segnali in ingresso al cv-x(2)




Figura 5.3: La pagina del sistema di visione

Infine con il pulsante "Calibra"(3) è possibile eseguire la procedura di calibrazione per l'aggiornamento dei dati del sistema di riferimento della camera.

5.3 La pagina delle segnalazioni

Accedendo alla pagina delle segnalazioni, premendo lapposito tasto precedentemente descritto, è possibile visualizzare tutte le segnalazioni dovute ad anomalie delle unità del sistema.

Le segnalazioni di allarme possono essere di due tipologie, con possibilità di ripristino automatico e con ripristino manuale. Tutte le segnalazioni abbinate al ripristino automatico scompaiono non appena la causa scatenante dellallarme viene risolta, mentre le segnalazioni con ripristino manuale vengono memorizzate per essere necessariamente visionate e riconosciute dalloperatore: tutti i movimenti legati allallarme in corso rimangono pertanto congelati in modo da consentire un eventuale sopralluogo; in questo caso ciclo riprenderà solo dopo il reset dell'allarme.




Figura 5.4: La pagina delle segnalazioni

In Figura 5.4 è mostrata una segnalazione di arresto del robot causata dalla mancanza di prodotti nell'area di prelievo.

Per effettuare il reset della segnalazione, è necessario premere il tasto di "Reset Allarmi" nella Home page.

5.4 La pagina delle ricette

Tramite la pagina di gestione delle ricette è possibile selezionare la ricetta in editazione(1) ed modificarne i dati. Premendo il pulsante SALVA(2) è possibile salvare nell'archivio i dati della ricetta in editazione. Il pulsante ATTIVA(3) apre il popup di attivazione di una nuova ricetta mentre il pulsante COPIA(4) alla sua destra apre il popup di copiatura ricette.




Figura $5.5\colon$ La pagina delle ricette

5.5 La pagina WorldZone




Figura 5.6: La pagina WorldZone

Dalla vista WorldZones è possibile visualizzare la posizione attuale del TCP del robot all'interno del layout. L'area è divisa nell'area di prelievo, deposito, home, magazzino e area al di fuori dell'area di lavoro.

5.6 La pagina dei dati macchina

Lambiente di setup è accessibile premendo lapposito tasto presente sulla home page del pannello. Tramite questa pagina è possibile modificare i dati fissi della macchina. nel caso specifico di questa applicazione l'unico dato issa consiste nella velocità del robot, selezionabile dalla dallo slider dai tasti alle sue estremità




Figura 5.7: La pagina dei dati macchina

Capitolo 6

Conclusioni

In questa trattazione è stato analizzato lo sviluppo di una tipica applicazione di automazione industriale. Si è quindi discusso il linguaggio di programmazione RAPID per i robot, evidenziandone gli aspetti di maggiore interesse e le particolarità rispetto ad altre marche di robot. Dopo un breve accenno al sistema di visione a alle difficoltà affrontate durante il suo utilizzo si è descritto il software lato ple e la comunicazione tra le diverse unità.

È stato infine discusso il funzionamento del pannello per permettere la comunicazione tra l'operatore e il sistema. Nello sviluppo si è cercato di proporre una soluzione che fosse il più possibile semplice ed intuitiva per l'operatore.

L'applicazione sviluppata si apre a diversi sviluppi futuri come la definizione di un numero superiore di formati e un ottimizzazione del tempo ciclo.

Acronyms

 $\mathbf{GSDML}\,$ General Station Description Markup Language.

HMI Human Machine Interface.

 ${\bf PLC}\,$ Programmable Logic Controller.

TCP Tool Center Point.

Bibliografia

- [1] Profinet: lo standard di comunicazione sicuro nelle reti industriali.
- $[2] \ ABB. \ IRC5 \ Industrial \ Robot \ Controller \ data-sheet.$
- [3] ABB. Manuale tecnico di riferimento: Istruzioni RAPID, Funzioni e Tipi di dati.
- [4] ABB. Technical reference manual RAPID kernel.
- [5] Siemens. Structured Control Language (SCL) for S7-300/S7-400 Programming.